MATEMATICKA ANALYZA V MAPLU

Symbolicke derivovani

```
> restart;
```

Pomoci procedury diff muzeme derivovat formule (vyrazy):

```
> 'diff(exp(-x^2),x)';
```

$$\frac{d}{dx}\left(e^{(-x^2)}\right)$$

Apostrofy kolem predchazejiciho vyrazu zamezi vyhodnoceni.
Stejneho efektu dosahneme i procedurou Diff. Diff se pouziva pro vetsi prehlednost a z duvodu kontroly spravnosti zadani.

```
> %;
```

$$-2\,x\,e^{(-x^2)}$$

```
> Diff(ln(x/(x^2+1)),x):%=value(%);
```

$$\frac{d}{dx}\ln\left(\frac{x}{x^2+1}\right)=\frac{\left(\frac{1}{x^2+1}-\frac{2\,x^2}{(x^2+1)^2}\right)(x^2+1)}{x}$$

```
> normal(%);
```

$$\frac{d}{dx}\ln\left(\frac{x}{x^2+1}\right)=-\frac{x^2-1}{x\,(x^2+1)}$$

```
> Diff(x^(x^x),x):%=value(%);
```

$$\frac{d}{dx}x^{(x^x)}=x^{(x^x)}\left(x^x\,(\ln(x)+1)\ln(x)+\frac{x^x}{x}\right)$$

```
> collect(%,ln(x), simplify); #diva se na vyraz jako na polynom v
  promenne ln(x)
```

$$\frac{d}{dx}x^{(x^x)}=x^{(x^x+x)}\ln(x)^2+x^{(x^x+x)}\ln(x)+x^{(x^x+x-1)}$$

Derivace vyssich radu:

```
> Diff(exp(-x^2),x,x):%=value(%);
```

$$\frac{d^2}{dx^2}\left(e^{(-x^2)}\right)=-2\,e^{(-x^2)}+4\,x^2\,e^{(-x^2)}$$

```
> Diff(exp(-x^2), x$5):%=value(%);
```

$$\frac{d^5}{dx^5}\left(e^{(-x^2)}\right) = -120\, x\, e^{(-x^2)} + 160\, x^3\, e^{(-x^2)} - 32\, x^5\, e^{(-x^2)}$$

Derivace funkce dane implicitne

```
> restart;
> alias(y=y(x)): #y povazujeme za funkci x
> eq:=x^2+y^2=c;
```

$$eq := x^2 + y^2 = c$$

```
> diff(eq,x);
```

$$2\, x + 2\, y \left(\frac{\partial}{\partial x} y\right) = 0$$

```
> dydx:=solve(%, diff(y,x)); # 1. derivace
```

$$dydx := -\frac{x}{y}$$

```
> diff(eq,x$2);
```

$$2 + 2 \left(\frac{\partial}{\partial x} y\right)^2 + 2\, y \left(\frac{\partial^2}{\partial x^2} y\right) = 0$$

```
> solve(%,diff(y,x$2));
```

$$-\frac{1 + \left(\frac{\partial}{\partial x} y\right)^2}{y}$$

```
> d2ydx2:=normal(subs(diff(y,x)=dydx,%));
```

$$d2ydx2 := -\frac{x^2 + y^2}{y^3}$$

```
> alias(y=y):
> restart;
> implicitdiff(x^2+y^2,y,x,x);
```

$$-\frac{x^2 + y^2}{y^3}$$

Parcialni derivace:

```
> Diff(exp(a*x*y^2),x,y$2):%=value(%);
```

$$\frac{\partial^3}{\partial y^2\, \partial x}\left(e^{(a\, x\, y^2)}\right) = 2\, a\, e^{(a\, x\, y^2)} + 10\, a^2\, y^2\, x\, e^{(a\, x\, y^2)} + 4\, a^3\, y^4\, x^2\, e^{(a\, x\, y^2)}$$

```
> factor(%);
```

$$\frac{\partial^3}{\partial y^2\,\partial x}\left(e^{(a\,x\,y^2)}\right) = 2\,a\,e^{(a\,x\,y^2)}\,(1 + 5\,a\,x\,y^2 + 2\,a^2\,y^4\,x^2)$$

```
> Diff(sin(x+y)/y^4, x$5, y$2):%=value(%);
```

$$\frac{\partial^7}{\partial y^2\,\partial x^5}\left(\frac{\sin(x+y)}{y^4}\right) = -\frac{\cos(x+y)}{y^4} + \frac{8\,\sin(x+y)}{y^5} + \frac{20\,\cos(x+y)}{y^6}$$

```
> collect(%,cos(x+y),normal);
```

$$\frac{\partial^7}{\partial y^2\,\partial x^5}\left(\frac{\sin(x+y)}{y^4}\right) = -\frac{(y^2-20)\,\cos(x+y)}{y^6} + \frac{8\,\sin(x+y)}{y^5}$$

Pokud derivuje funkci (ve smyslu datove struktury Maplu), musime pouzit funkcniho operatoru D.

```
> g:=x->x^n*exp(sin(x));
```

$$g := x \to x^n\,e^{\sin(x)}$$

```
> D(g);
```

$$x \to \frac{x^n\,n\,e^{\sin(x)}}{x} + x^n\,\cos(x)\,e^{\sin(x)}$$

```
> D(g)(Pi/6);
```

$$\frac{6\left(\frac{\pi}{6}\right)^n n\,e^{(1/2)}}{\pi} + \frac{1}{2}\left(\frac{\pi}{6}\right)^n \sqrt{3}\,e^{(1/2)}$$

diff derivuje vzorec a na vystupu vraci vzorec, D derivuje funkci a na vystupu vraci funkci. Priklady:

```
> Eval(Diff(g(x), x), x=Pi/6);
```

$$\left.\left(\frac{\partial}{\partial x}\left(x^n\,e^{\sin(x)}\right)\right)\right|_{x=\frac{\pi}{6}}$$

```
> value(%);
```

$$\frac{6\left(\frac{\pi}{6}\right)^n n\,e^{(1/2)}}{\pi} + \frac{1}{2}\left(\frac{\pi}{6}\right)^n \sqrt{3}\,e^{(1/2)}$$

```
> diff(cos(t),t); #derivace vzorce
```

$$-\sin(t)$$

```
> D(cos); #derivace funkce
```

$$-\sin$$

```
> (D@@2)(cos); #pro druhou derivaci funkce musime pouzit operatoru
```

```
skladani funkci
```

$$-\cos$$

```
> D(cos)(t); # derivace funkce v danem bode
```

$$-\sin(t)$$

Vsimnete si rozdilu mezi nasledujicimi dvema prikazy:

```
> D(cos(t));
```

$$D(\cos(t))$$

Maple povazuje cos(t) za slozeni funkci cos a t, spravny zapis je tedy:

```
> D(cos @ t);
```

$$((-\sin)@t)\,D(t)$$

Derivace implicitni funkce pomoci operatoru D:

```
> eq:=x^2+y^2=c:
> D(eq);
```

$$2\,D(x)\,x + 2\,D(y)\,y = D(c)$$

```
> solve(%,D(y));
```

$$-\frac{1}{2}\,\frac{2\,D(x)\,x - D(c)}{y}$$

```
> dydx:=subs(D(x)=1, D(c)=0, %);
```

$$dydx := -\frac{x}{y}$$

```
> (D@@2)(eq);
```

$$2\,(D^{(2)})(x)\,x + 2\,D(x)^2 + 2\,(D^{(2)})(y)\,y + 2\,D(y)^2 = (D^{(2)})(c)$$

```
> solve(%,(D@@2)(y));
```

$$-\frac{1}{2}\,\frac{2\,(D^{(2)})(x)\,x + 2\,D(x)^2 + 2\,D(y)^2 - (D^{(2)})(c)}{y}$$

```
> d2ydx2:=normal(subs(D(x)=1, (D@@2)(x)=0, (D@@2)(c)=0, D(y)=dydx,
  %));
```

$$d2ydx2 := -\frac{x^2 + y^2}{y^3}$$

Operatoru D je mozno pouzit i pro vypocet parcialnich derivaci:

```
> h:=(x,y,z)->1/(x^2+y^2+z^2)^(1/2);
```

$$h := (x, y, z) \rightarrow \frac{1}{\sqrt{x^2+y^2+z^2}}$$

```
> 'D[1](h)'=D[1](h);
```

$$D_1(h) = \left((x, y, z) \rightarrow -\frac{x}{(x^2+y^2+z^2)^{(3/2)}}\right)$$

Zde D[1](h) je parcialni derivace vzhledem k x.

```
> 'D[1,2](h)'=D[1,2](h);
```

$$D_{1,2}(h) = \left((x, y, z) \rightarrow \frac{3\,y\,x}{(x^2+y^2+z^2)^{(5/2)}}\right)$$

D[1,2](h) je vlastne D[1](D[2](h)) - smisena parcialni derivace vzhledem k x a y.

```
> 'D[1,1](h)'=D[1,1](h);
```

$$D_{1,1}(h) = \left((x, y, z) \rightarrow \frac{3\,x^2}{(x^2+y^2+z^2)^{(5/2)}} - \frac{1}{(x^2+y^2+z^2)^{(3/2)}}\right)$$

Druha parcialni derivace vzhledem k x.

```
> L[h]:=(D[1,1]+D[2,2]+D[3,3])(h);
```

$$L_h := \left((x, y, z) \rightarrow \frac{3\,x^2}{(x^2+y^2+z^2)^{(5/2)}} - \frac{1}{(x^2+y^2+z^2)^{(3/2)}}\right) + \left((x, y, z) \rightarrow \frac{3\,y^2}{(x^2+y^2+z^2)^{(5/2)}} - \frac{1}{(x^2+y^2+z^2)^{(3/2)}}\right) + \left((x, y, z) \rightarrow \frac{3\,z^2}{(x^2+y^2+z^2)^{(5/2)}} - \frac{1}{(x^2+y^2+z^2)^{(3/2)}}\right)$$

```
> normal(L[h](x,y,y));
```

$$0$$

Maple muze derivovat i po castech definovane funkce:

```
> F:=x->piecewise(x>0, sin(x), arctan(x));
```

$$F := x \rightarrow \mathrm{piecewise}(0 < x, \sin(x), \arctan(x))$$

```
> Fp:=D(F);
```

$$Fp := x \rightarrow \mathrm{piecewise}\left(x \le 0, \frac{1}{1+x^2}, 0 < x, \cos(x)\right)$$

```
> plot({F, Fp}, -3*Pi..3*Pi);
```




```
> F:=x->if x>0 then sin(x) else arctan(x) fi;
```

F **:= proc(*x*) option *operator, arrow*; if 0 < *x* then sin(*x*) else arctan(*x*) end if end proc**

```
> Fp:=D(F);
```

Fp **:= proc(*x*) option *operator, arrow*; if 0 < *x* then cos(*x*) else 1 / (1 + *x*^2) end if end proc**

```
> plot({F, Fp}, -3*Pi..3*Pi);
```




```
> restart;
```

Integrace a sumace

Maple pouziva k integraci specialni algoritmy (napr. tzv. Rischuv algoritmus).

Procedura Int integral nevyhodnocuje, pouze prepisuje.

```
> Int(x/(x^3+1), x):%=value(%);
```

$$\int \frac{x}{x^3+1}\,dx = -\frac{1}{3}\ln(x+1)+\frac{1}{6}\ln(x^2-x+1)+\frac{1}{3}\sqrt{3}\,\arctan\left(\frac{(2\,x-1)\sqrt{3}}{3}\right)$$

```
> diff(rhs(%), x);
```

$$-\frac{1}{3\,(x+1)}+\frac{2\,x-1}{6\,(x^2-x+1)}+\frac{2}{3\left(1+\frac{(2\,x-1)^2}{3}\right)}$$

Pomoci rhs() se odkazujeme na pravou stranu rovnice, lhs() na levou.

```
> normal(%,’expanded’);
```

$$\frac{x}{x^3+1}$$

```
> infolevel[int]:=2:
> Int(x/(x^5+1),x): %=value(%);
```

```
int/indef1:   first-stage indefinite integration
int/ratpoly:   rational function integration
int/ratpoly:   rational function integration
```

$$\int \frac{x}{x^5+1}\,dx = -\frac{1}{5}\ln(x+1)+\frac{1}{20}\ln(-2x^2+x+\sqrt{5}\,x-2)-\frac{1}{20}\ln(-2x^2+x+\sqrt{5}\,x-2)\sqrt{5}$$
$$-\frac{2}{5}\,\frac{\arctan\left(\frac{-4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)\sqrt{5}}{\sqrt{10-2\sqrt{5}}}+\frac{1}{20}\ln(2x^2-x+\sqrt{5}\,x+2)\sqrt{5}+\frac{1}{20}\ln(2x^2-x+\sqrt{5}\,x+2)$$
$$-\frac{2}{5}\,\frac{\arctan\left(\frac{4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)\sqrt{5}}{\sqrt{10+2\sqrt{5}}}$$

```
> normal(diff(rhs(%),x), 'expanded');
```

$$\frac{x}{x^5+1}$$

```
> infolevel[int]:=0:
```

Maple pri vypoctech automaticky voli nulovou integracni konstantu.

```
> Int(2*x*(x^2+1)^24, x);
```

$$\int 2\,x\,(x^2+1)^{24}\,dx$$

```
> value(%);
```

$$208012\,x^{24}+81719\,x^{32}+7084\,x^{38}+19228\,x^{14}+19228\,x^{36}+\frac{653752}{5}\,x^{30}+7084\,x^{12}+43263\,x^{34}$$
$$+x^2+92\,x^{44}+506\,x^{42}+\frac{653752}{5}\,x^{20}+81719\,x^{18}+43263\,x^{16}+\frac{10626}{5}\,x^{40}+178296\,x^{22}$$
$$+208012\,x^{26}+\frac{1}{25}\,x^{50}+x^{48}+\frac{10626}{5}\,x^{10}+178296\,x^{28}+12\,x^{46}+12\,x^4+92\,x^6+506\,x^8$$

```
> factor(%);
```

$$x^2\,(x^8+5\,x^6+10\,x^4+10\,x^2+5)\,(x^{40}+20\,x^{38}+190\,x^{36}+1140\,x^{34}+4845\,x^{32}+15505\,x^{30}$$
$$+38775\,x^{28}+77625\,x^{26}+126425\,x^{24}+169325\,x^{22}+187760\,x^{20}+172975\,x^{18}+132450\,x^{16}$$
$$+84075\,x^{14}+43975\,x^{12}+18760\,x^{10}+6425\,x^8+1725\,x^6+350\,x^4+50\,x^2+5)\,/\,25$$

```
> factor(%+1/25);
```

$$\frac{(x^2+1)^{25}}{25}$$

```
> restart;
```

Urcity integral:

```
> Integrate(x/(x^3+1), x=1..2): %=value(%);
```

$$\int_{1}^{2} \frac{x}{x^3+1}\,dx = \frac{1}{3}\ln(2) + \frac{\sqrt{3}\,\pi}{18} - \frac{1}{6}\ln(3)$$

```
> Int(1/((1+x^2)*(1+2*x^2)), x=0..1):
> %=value(%);
```

$$\int_{0}^{1} \frac{1}{(1+x^2)(1+2\,x^2)}\,dx = -\frac{\pi}{4} + \sqrt{2}\,\arctan(\sqrt{2})$$

```
> Int(1/x^2, x=-1..1): %=value(%);
```

$$\int_{-1}^{1} \frac{1}{x^2}\,dx = \infty$$

```
> Int(1/x^2, x): %=value(%);
```

$$\int \frac{1}{x^2}\,dx = -\frac{1}{x}$$

```
> subs(x=1, rhs(%)) - subs(x=-1, rhs(%));
```

$$-2$$

```
> plot(1/x^2, x=-1..1, y=-10..10);
```



Maple kontroluje nespojitosti integrandu na zadanem intervelu.

Nevlastni integraly:

```
> Int(t^4*ln(t)^2/(1+3*t^2)^3, t=0..infinity): %=value(%);
```

$$\int_0^\infty \frac{t^4 \ln(t)^2}{(1+3\,t^2)^3}\,dt = \frac{\sqrt{3}\,\pi}{216} + \frac{\pi^3\sqrt{3}}{576} - \frac{1}{108}\,\pi\sqrt{3}\,\ln(3) + \frac{1}{576}\,\pi\sqrt{3}\,\ln(3)^2$$

V pripade, ze Maple neni schopen najit reseni symbolicky, muzeme pouzit numerickeho integrovani:

```
> int(exp(arcsin(x)), x=0..1);
```

$$\int_0^1 e^{\arcsin(x)}\,dx$$

```
> evalf(%);
```

$$1.905238690$$

Nekdy je vyhodne Maplu pri reseni asistovat (pokud chceme i postup reseni):

```
> with(student):
```

```
> Int(sqrt(9-x^2), x);
```

$$\int \sqrt{9 - x^2}\, dx$$

```
> changevar(x=3*sin(t), Int(sqrt(9-x^2), x),t);
```

$$\int 3\sqrt{9 - 9\sin(t)^2}\cos(t)\, dt$$

```
> value(%);
```

$$\frac{9}{2}\sqrt{1 - \sin(t)^2}\sin(t) + \frac{9\,t}{2}$$

```
> simplify(subs(t=arcsin(x/3), %));
```

$$\frac{\sqrt{9 - x^2}\, x}{2} + \frac{9}{2}\arcsin\left(\frac{x}{3}\right)$$

Pro metodu per - partes:

```
> i:=Int((x^2+1)*ln(x), x);
```

$$i := \int (x^2 + 1)\ln(x)\, dx$$

```
> i=intparts(i, ln(x));
```

$$\int (x^2 + 1)\ln(x)\, dx = \ln(x)\left(x + \frac{1}{3}x^3\right) - \int \frac{x + \frac{1}{3}x^3}{x}\, dx$$

```
> i=value(lhs(%));
```

$$\int (x^2 + 1)\ln(x)\, dx = x\ln(x) - x + \frac{1}{3}x^3\ln(x) - \frac{x^3}{9}$$

Zobrazeni postupu vypoctu:

```
> read "VypocetIntegralu.txt":
> VypocetIntegralu(exp(x)*sin(x), czech);
```

$$\int e^x \sin(x)\, dx =$$

[*u ijeme metodu per partes*, $u = e^x$, $v = -\cos(x)$]

$$= -e^x\cos(x) - \int -e^x\cos(x)\, dx$$

[*u ijeme metodu per partes*, $u = e^x$, $v = \sin(x)$]

$$= -e^x\cos(x) + e^x\sin(x) - \int e^x\sin(x)\, dx$$

[*integrºl spoµt°me algebraicky*]

$$= -\frac{1}{2} e^x \cos(x) + \frac{1}{2} e^x \sin(x)$$

[*uprav¾me v raz*]

$$= \frac{1}{2} e^x (-\cos(x) + \sin(x))$$

```
> with(Student[Calculus1]):
> IntTutor();
Initializing Java runtime environment.
```

$$\int e^x \sin(x)\, dx = -\frac{1}{2} e^x \cos(x) + \frac{1}{2} e^x \sin(x)$$

```
> Int(exp(x)*sin(x), x);
```

$$\int e^x \sin(x)\, dx$$

```
> Hint(%);
```

$$[parts, e^x, -\cos(x)]$$

```
> restart;
```

Konecne a nekonecne soucty:

```
> Sum(k^7, k=1..20): %=value(%);
```

$$\sum_{k=1}^{20} k^7 = 3877286700$$

```
> Sum(k^7, k=1..n): %=value(%);
```

$$\sum_{k=1}^{n} k^7 = \frac{(n+1)^8}{8} - \frac{(n+1)^7}{2} + \frac{7\,(n+1)^6}{12} - \frac{7\,(n+1)^4}{24} + \frac{(n+1)^2}{12}$$

```
> factor(%);
```

$$\sum_{k=1}^{n} k^7 = \frac{n^2\,(3\,n^4 + 6\,n^3 - n^2 - 4\,n + 2)\,(n+1)^2}{24}$$

```
> Sum(1/(k^2-4), k=3..infinity): %=value(%);
```

$$\sum_{k=3}^{\infty} \frac{1}{k^2-4} = \frac{25}{48}$$

Tayloruv rozvoj

```
> taylor(sin(tan(x))-tan(sin(x)), x=0, 25);
```

$$-\frac{1}{30} x^7 - \frac{29}{756} x^9 - \frac{1913}{75600} x^{11} - \frac{95}{7392} x^{13} - \frac{311148869}{54486432000} x^{15} - \frac{10193207}{4358914560} x^{17} - \frac{1664108363}{1905468364800}$$

$$x^{19} - \frac{2097555460001}{7602818775552000} x^{21} - \frac{374694625074883}{6690480522485760000} x^{23} + O(x^{25})$$

```
> whattype(%);
```

$$series$$

```
> order(%%);
```

$$25$$

```
> 25;
```

$$25$$

```
> Order;
```

$$6$$

```
> Order:=3: taylor(f(x), x=a);
```

$$f(a) + D(f)(a)(x-a) + \frac{1}{2}(D^{(2)})(f)(a)(x-a)^2 + O((x-a)^3)$$

```
> sin_series:=taylor(sin(x), x=0, 6);
```

$$sin\_series := x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^7)$$

```
> sin_series := series(1*x-1/6*x^3+1/120*x^5+O(x^7),x,7);
```

$$sin\_series := x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^7)$$

I kdyz struktura rozvoje nam pripomina polynom, interni datova reprezentace je jina:

```
> subs(x=2, sin_series);

Error, invalid substitution in series
```

```
> op(sin_series);
```

$$1, 1, \frac{-1}{6}, 3, \frac{1}{120}, 5, O(1), 7$$

```
> sin_series*sin_series;
```

$$\left(x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^7)\right)^2$$

```
> expand(%);
```

$$\left(x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^7)\right)^2$$

```
> taylor(%,x);
```

$$x^2 + O(x^4)$$

```
> mtaylor(sin(x), x=0,6);
```

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5$$

mtaylor pocita Taylorovy rozvoje i pro funkce vice promennych a vysledkem je datova struktura typu polynom.

```
> subs(x=2, %);
```

$$\frac{14}{15}$$

```
> whattype(%%);
```

$$+$$

Muzeme urcovat i koeficienty u danych mocnin x bez nutnosti pocitat cely rozvoj:

```
> coeftayl(sin(x), x=0, 19);
```

$$\frac{-1}{121645100408832000}$$

```
> sin_series;
```

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^7)$$

```
> diff(sin_series, x);
```

$$1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 + O(x^6)$$

```
> integrate(sin_series, x);
```

$$\frac{1}{2}x^2 - \frac{1}{24}x^4 + \frac{1}{720}x^6 + O(x^8)$$

```
> convert(sin_series, 'polynom');
```

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5$$

```
> restart;
> with(powseries); #procedury, slouzici k praci s mocninnymi
  radami
```

[*compose, evalpow, inverse, multconst, multiply, negative, powadd, powcos, powcreate, powdiff, powexp, powint, powlog, powpoly, powsin, powsolve, powsqrt, quotient, reversion, subtract, template, tpsform*]

```
> powcreate(f(n)=a^n/n!);
```

```
> powcreate(g(n)=(-1)^(n+1)/n,g(0)=0);
> f(2);
```

$$\frac{a^2}{2}$$

```
> f_series:=tpsform(f,x,5);
```

$$f\_series := 1 + a\,x + \frac{a^2}{2}\,x^2 + \frac{a^3}{6}\,x^3 + \frac{a^4}{24}\,x^4 + \mathrm{O}(x^5)$$

```
> g_series:=tpsform(g,x,5);
```

$$g\_series := x - \frac{1}{2}\,x^2 + \frac{1}{3}\,x^3 - \frac{1}{4}\,x^4 + \mathrm{O}(x^5)$$

```
> s:=powadd(f,g): tpsform(s,x,3);
```

$$1 + (a+1)\,x + \left(\frac{a^2}{2} - \frac{1}{2}\right)x^2 + \mathrm{O}(x^3)$$

```
> series(1+(a+1)*x+(1/2*a^2-1/2)*x^2+O(x^3),x,3);
```

$$1 + (a+1)\,x + \left(\frac{a^2}{2} - \frac{1}{2}\right)x^2 + \mathrm{O}(x^3)$$

```
> p:=multiply(f,g): tpsform(p,x,4);
```

$$x + \left(-\frac{1}{2} + a\right)x^2 + \left(\frac{1}{3} - \frac{1}{2}\,a + \frac{1}{2}\,a^2\right)x^3 + \mathrm{O}(x^4)$$

```
> d:=powdiff(f): tpsform(d,x,4);
```

$$a + a^2\,x + \frac{a^3}{2}\,x^2 + \frac{a^4}{6}\,x^3 + \mathrm{O}(x^4)$$

```
> i:=powint(f): tpsform(i,x,4);
```

$$x + \frac{a}{2}\,x^2 + \frac{a^2}{6}\,x^3 + \mathrm{O}(x^4)$$

Vypocty limit

```
> Limit( cos(x)^(1/x^3), x=0): %=value(%);
```

$$\lim_{x \to 0} \cos(x)^{\left(\frac{1}{x^3}\right)} = undefined$$

Jednostranne limity:

```
> Limit(cos(x)^(1/x^3), x=0, 'right'):
> %=value(%);
```

$$\lim_{x \to 0+} \cos(x)^{\left(\frac{1}{x^3}\right)} = 0$$

```
> Limit(cos(x)^(1/x^3), x=0, 'left'):
> %=value(%);
```

$$\lim_{x \to 0-} \cos(x)^{\left(\frac{1}{x^3}\right)} = \infty$$

```
> y:=exp(a*x)*cos(b*x);
```

$$y := e^{(a\,x)} \cos(b\,x)$$

```
> limit(y, x=-infinity);
```

$$\lim_{x \to (-\infty)} e^{(a\,x)} \cos(b\,x)$$

```
> assume(a>0):
> limit(y, x=-infinity);
```

$$0$$

```
> limit(y, x=-infinity) assuming a>0;
```

$$0$$

Student Calculus Package

```
> restart;
> with(student):
> f:=x->-2/3*x^2+x;
```

$$f := x \to -\frac{2}{3}x^2 + x$$

```
> (f(x+h)-f(x))/h;
```

$$\frac{-\frac{2\,(x+h)^2}{3} + h + \frac{2\,x^2}{3}}{h}$$

```
> Limit(%,h=0);
```

$$\lim_{h \to 0} \frac{-\frac{2\,(x+h)^2}{3} + h + \frac{2\,x^2}{3}}{h}$$

```
> value(%);
```

$$-\frac{4x}{3}+1$$

```
> eval(%, x=0);
```

$$1$$

```
> showtangent(f(x), x=0);
```




```
> intercept(y=f(x), y=0);
```

$$\{y=0, x=0\}, \{y=0, x=\frac{3}{2}\}$$

```
> middlebox(f(x), x=0..3/2);
```




```
> middlebox(f(x), x=0..3/2, 10);
```




```
> middlesum(f(x), x=0..3/2, 10);
```

$$\frac{3}{20}\left(\sum_{i=0}^{9}\left(-\frac{2\left(\frac{3\,i}{20}+\frac{3}{40}\right)^2}{3}+\frac{3\,i}{20}+\frac{3}{40}\right)\right)$$

```
> value(%);
```

$$\frac{603}{1600}$$

```
> middlesum(f(x), x=0..3/2, n);
```

$$\frac{3}{2}\,\frac{\sum_{i=0}^{n-1}\left(-\frac{3\left(i+\frac{1}{2}\right)^2}{2\,n^2}+\frac{3\left(i+\frac{1}{2}\right)}{2\,n}\right)}{n}$$

```
> Limit(%, n=infinity);
```

$$\lim_{n\to\infty} \frac{3}{2} \frac{\displaystyle\sum_{i=0}^{n-1}\left(-\frac{3\left(i+\frac{1}{2}\right)^2}{2\,n^2}+\frac{3\left(i+\frac{1}{2}\right)}{2\,n}\right)}{n}$$

```
> value(%);
```

$$\frac{3}{8}$$

```
> Int(f(x), x=0..3/2);
```

$$\int_0^{3/2} -\frac{2}{3}\,x^2 + x\,dx$$

```
> value(%);
```

$$\frac{3}{8}$$

```
> with(Student[Calculus1]):
> DiffTutor();
```

$$\frac{d}{dx}\left(x\sin(x)\right) = \sin(x) + x\cos(x)$$

```
> IntTutor();
Initializing Java runtime environment.
```

$$\int x^2\sin(x)\,dx = -x^2\cos(x) + 2\,x\sin(x) + 2\cos(x)$$

```
>
```